United States

Department of Interior

Geological Survey

FABSTAT:

A Card-imaged FORTRAN Program to be used for Axially distributed Planar-Structural Data

by

William J. Perry, Jr.

Open-file Report 81-158

1981

FABSTAT:

A Card-imaged FORTRAN Program to be used for

Axially distributed Planar-Structural Data

CONTENTS

INTRODUCTION

Fabstat is an acronym for fabric statistics. "fabstat.fortran" is a card-imaged Multics FORTRAN program that treats uniaxial structural fabrics in terms of hemispherical distributions of unit vectors. Means, radii of confidence and angles of error of the means are calculated at p=.05 (95% confidence level) for up to nine sets of planes per locality, using Fisher's (1953) "R test" for spherical distributions.

The mean calculated by this program is the vector sum of the set of unit vectors which represent the poles to planes of the respective data set, divided by r, the length of the vector sum, such that the vector mean is a unit vector. The radius of confidence R is the cosine of the angle of error, the probable error of the mean pole computed at the given confidence level. Within this program, the probability is set at 5% that the computed mean will lie more than X degrees away from the true mean where X represents the angle of error. For values of r that are at least 2 less than the number of data planes in the set, the computation of R is inaccurate (Fisher, 1953). In this case the program will not compute R but will print an error message.

The poles to the mean planes and data planes of the respective sets of data are rotated to bedding flat by the program if the strike and dip are entered according to appropriate conventions. As presently dimensioned, this program will handle as many as 60 data planes per locality.

As many localities as desired can be handled by one run of the program, as each locality represents a separate data packet, and data packets coupled together (using proper input conventions) will run consecutively without mutual interference. Line length of this text is set at 80 spaces to match the 80 columns (spaces) of card-image FORTRAN, so that the source code is given in compilable form. All examples of fixed-format input given in this report are in appropriate columns. This format must be followed for meaningful output.

The program as reproduced below has run successfully on the Honeywell Multics system as recently as January, 1981, and was used extensively during 1980 to analyse joint patterns in the Appalachian Basin using input option 1, described below.

DESCRIPTION

In order to provide sufficient background to the geologist using this program, the source code (below) is extensively documented with comments. This documentation will not be repeated here, only summarized. The fabstat.fortran program is subdivided into the following sections:

(1) Explanation of terminology.
(2) Last modification date and input/output conventions.
(3) Dimensions and internal definitions
(4) Reading of data and initial sorting operations.
(5) Initial write statements and computations.
(6) Determination of mean poles to planes and angles of error.
(7) Determination of rotation matrix.
(8) Rotation of data to bedding flat.
(9) Final sorting and write statements.
(10) Subroutine mean, which computes means and radii of confidence.
(11) Subroutine angle, which computes azimuth and plunge.
(12) Subroutine unvec, which verifies unit vector arrays.

Sections 7 through 9 operate only if non-zero values for strike and dip of bedding are entered in line 4 (card 4) of the data segment, as shown on page 2. The rotation matrix and other matrices and arrays are defined in section one. The working part of the program begins with the FORTRAN dimension statement, which defines the amount of data which can be stored in the various arrays and matrices: as many as sixty data planes in as many as nine sets per locality.

Two conventions are followed which may cause problems on systems other than Multics:
First, "character*n" statements are used to define maximum word lengths of character strings (an ANSI Fortran 77 convention); second the free-format statement "format (v)" is used for list-directed input and output (a Multics FORTRAN convention). To make the second conform with Fortran 77, the format statement is omitted, and the read or write statements changed to the form "read (n,*)" or "write (n,*)". For older compilers this program would need to be substantially modified. For systems other than Multics, all characters in the source code may need to be capitalized.

The interested reader is referred to the source code (p.4-13) for further details. The program is set up for maximum flexibility without being interactive. The next section gives instructions for entering the data. The following section (page 3) shows one method of attaching the data package to the program, running the program and getting useable output.

INPUT CONVENTIONS

For meaningful output, it is critically important that the data be arranged properly. Sample data segments are given in the appendix (pages 15 and 24) with sample output. For input option 1 and in all output, strikes are given in azimuth form (0 to 359 degrees). In order to uniquely define dip direction, the strike azimuth must be given as the value 90 degrees counterclockwise from the dip direction, i.e., if the strike is N 45 degrees W and the dip 78 degrees SW using classical geologic conventions, then for input option 1 and in all output this orientation will appear as "135. 78.".

The only exceptions are described under option 1 in the source code (below). If the data are to be punched on cards, line numbers (below) are to be considered card numbers. Column numbers (below) refer to the standard 80 spaces of a computer card, numbered from left to right.

The first line of data gives the option number in column 1:
1 (if option 1 chosen as described above and again more thoroughly in the source code).
2 (if classical geologic conventions are to be used in input, i.e., N 80.W, 80.SW; described more thoroughly under input option 2 in the source code).

The second line provides general geographic and geologic character-string information concerning the locality. This line is free-formatted, and the operator must enter all five character strings separated by single blanks with no internal blanks: first, the locality title (9 characters maximum), second, the 1dx2d sheet (16 characters maximum), third, the county (16 characters maximum), fourth, the quadrangle (16 characters maximum), and last, the system/formation at the locality (20 characters maximum). If any of these are unknown then type the word 'unknown' in the appropriate spot on the line.

The third line gives the latitude, longitude, and the date that the data were acquired, using the same conventions as above as this line is also free-formatted. Maximum character-string lengths for line 3 are all 9 characters.

Line 4 provides the mean bedding orientation at the locality. This data needs to be entered very differently, depending in your input option (1 or 2). If you are using option 1, this line of data is handled by format statement 303, which expects decimal points in columns 4 and 9. For example:

```
123456789 = column numbers.
045.5 25.   (azimuth in columns 1 through 5; dip in columns 6 through 9).
```

For overturned beds see comments in source-code listing under Exception 2, option 1. If option 2, this line of data is handled by format statement 310.

```
123456789 = column numbers.  For example:
N45.5E 25.SE _  (all in capital letters, using standard geologic abbreviations)
```

If instead of a blank or nothing in column 14, a capital letter O is typed, the program will interpret the bedding as overturned.

The fifth line gives the number of data sets input from the locality (up to 9) in column 1. The sixth line gives the lithology, rank, and number of measurements in set 1 and is free-formatted as discussed above. Lithology and rank may consist of as many as 12 characters each.

The seventh line and similar succeeding lines give the strike and dip of the data planes, three to a line, using the proper option conventions. Option 1 uses format statement 306; data must be entered in the form:

```
123456789 = column numbers.
173.103. 169. 87. 117. 87.  (see source code, option 1, Exception 1).
```

with decimal points in columns 4,8,13,17,22, and 26. If the decimal points are omitted, all numbers will need to be right justified one space. Otherwise for example, "87." will be read as "870". Option 2 uses format statement 315; data should be entered in the form:

```
S 7.E 83. W S11.E 87. W S10.E 87. W   (all letters capitalized, and everything spaced as shown).
```

Line 6 may be termed the data-set designator, whereas lines 7 and succeeding similar lines are the data set for set 1. Immediately following the data set for set 1 must be placed the data-set designator for set 2 followed by its data set if present. This is repeated for as many as ten data sets as stated above. No internal blank lines can be placed within or between any of the data; otherwise the program will not run properly.

Immediately following the last data set, is the declination-correction designator. On this line, the conventions are as follows:

123456789 = column numbers.

0.   (no declination corrections required, all data oriented to true north)

1. 3.5   (declination correction required; compass was set on 0 declination, whereas actual declination was 3.5 degrees west)

1.-17.   (declination correction required; compass was set on 0 declination, whereas actual declination was 17 degrees east)

The correction is given in columns 3 through 7 in the form ± __.__ degrees.

The final line of data must contain a 0 in column 1, unless another locality follows:

0   (tells the computer to stop; that this is the end of the last locality to be considered).

Otherwise any number 1 through 9 will be entered indicating as follows:

1   (another data package will follow with a 1 or 2 in column 1 on the following line, depending on the input option).

Maximum flexibility is provided to the user of this program to the extent that he/she specifies the number of localities, number of data sets within each locality, and number of data planes within each set to be considered in each run.

INTERFACING WITH MULTICS

The fabstat.fortran program may be run on the Multics system using an interfacing series of commands executed by calling fabstat.ec:

```
io attach file10 vfile_ rotate_data
io attach file11 vfile_ rotate_out
set_cc file11 -on
answer yes fabstat
io detach file(10 11)
qx fabstat
dp -dl RESULTS
```

The first command attaches file 10 of the program source code to a Multics segment "rotate_data" which must contain the data. The second command attaches file 11 of the program to a Multics segment "rotate_out" which will contain the output. The third command sets the carriage control on for file 11 so that the output will be formatted in the intended manner. The fourth command runs the program and answers yes to any question that comes up, so that the run can take place on an absentee basis. The next command detaches files 10 and 11. The "qx fabstat" command invokes fabstat.qedx, an editing subroutine which stacks the data segment and the output segment together in a temporary segment called "RESULTS". The final command tells the line printer to print RESULTS, after which the segment RESULTS is deleted.

```
c          **************************************************
c          *  FABSTAT.FORTRAN SOURCE CODE, CARD-IMAGE FORMAT  *
c          **************************************************
c
c                     EXPLANATION OF TERMINOLOGY
c
c
c    MAIN ARRAYS AND MATRICES:
c
c    alpha(n) - Strikes of planes. See input options 1 and 2 below.
c    beta(n) - Dips of planes from horizontal. See input options 1 and 2 below.
c
c    c(i,60) - direction cosine matrix of poles to planes designated by
c       alpha and beta. Convention: "i" varies from 1 to 3, 1 - north (on x
c       axis, 2 - east (on y axis), and 3 - down (on z axis).
c
c    cm(70,i) - direction cosine matrix of mean poles, "i" as above. Matrix
c       used in mid-program to store all data to be rotated to bedding flat.
c
c=============================================================================
c    CHARACTER ARRAYS:
c
c    lithology(9) - the lithology of sets 1 through 9.
c    rank(9) - the relative ranking of sets 1 through 9.
c
c==============================================================================
c    ROTATION MATRIX:
c
c    a(i,j) - derived from "al", the strike of bedding, and "be", the dip
c        of bedding. By matrix multiplication [a(i,j)x cm(70,j)] the coord-
c        inate axes are changed such that the data are rotated to bedding
c        flat.
c
c==============================================================================
```

```
c=========================================================================
c               PROGRAM - - - fabstat.fortran - - - August, 1980.
c                    Last modified 1/14/81 by W.J.Perry,Jr.
c
c As many as 9 sets of data from one locality may be entered from terminal or
c data deck, so long as unit 10 is attached to a formatted sequential input
c file. Makeup of sets are determined by field observer or interpreter.
c    The vector mean and angle of error about the mean are calculated
c for each set, mapping Fisher's spherical distribution onto a hemispherical
c distribution of data. Output goes to unit 11 which needs to be attached to
c an output file for printing, punching or scanning.  To convert this program
c to standard FORTRAN, convert lower case letters to upper case. This program
c has been compiled using the ansi66 Multics FORTRAN compiler with the '-card'
c and '-optimizer' arguments.
c*************************************************************************
c    DIMENSIONS AND INTERNAL DEFINITIONS:
c
      dimension alpha(70),beta(70),c(3,60),cm(70,3),prod(3),q(3),u(3),
     2a(3,3),lithology(9),rank(9),final(9),init(9),ddr(60),stdr1(60),
     3stdr2(60)
      common c,cm,alpha,beta,rad
      character*1 stdr1,stdr2,overt
      character*2 ddr
      character*9 lat,long,date,locality
      character*12 lithology, rank
      character*16 sheet,county ,quad
      character*20 systemfm
      integer final, option
      double precision rad
      rad=0.017453293 d0
c
c+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
c    READING OF DATA AND INITIAL SORTING OPERATIONS:
c
    5 read (10,300) option
c
c If option=1, strikes (alpha,al) are input as azimuths (up to 359 degrees
c    clockwise from north, using convention that dip direction is 90 degrees
c    clockwise from given strike; otherwise data will not be handled properly.
c  Exception 1: For steeply dipping planes within a set (alpha(n),beta(n)),
c    those in minority dip direction must be input as dip (beta) = 180 degrees
c    minus field-recorded dip; alpha must be input as azimuth, 90 degrees
c    clockwise from field-recorded dip direction, i.e., S40E, 85SW becomes
c    320. 95. if the majority dip direction is northeast.
c  Exception 2: For overturned strata represented by "al"(strike) and "be"
c    (dip),"al" is dip direction + 90 degrees and "be" is 180 degrees minus
c    the dip amplitude: 225. 105. (same as N45.E 75.SE 0 in option 2 below).
c If option=2, strike and dip are entered in quadrant manner, i.e., "N45.E",
c    " 85.SE". For those planes in minority dip direction (as exception 1
c    above), strike is given in same sector as remainder of data, i.e.,
c    "n ___.E", with dip input as exception 1 above.
```

```
c    For overturned bedding,"al,be" is  followed by capital letter "O" in
c    the second column after the dip direction: N45. E 75.SE O. Note that
c    the strike of bedding is formatted differently (f4.1) than the strike
c    of the data planes (f3.0).
c
c   OPTION 1:
       if (option.eq.2) go to 15
       read (10,301) locality,sheet,county,quad,systemfm
c   Convert to ANSI Fortran 77 by changing to 'read (10,*) locality,' etc.
       read (10,302) lat,long,date
c   Convert to ANSI Fortran 77 by changing to 'read(10,*) lat,long,date'.
       read (10,303) al,be
       read (10,304) num
c   "num"=number of sets considered from locality
       n=0
       do 10 i=1,num
       read (10,305) lithology(i),rank(i),num_jts
c   Convert to ANSI Fortran 77 by changing to 'read (10,*)' etc.
c   "num_jts" is the number of joints in set "i".
       init(i)=n+1
       l=init(i)
c   "init(i)" is the first element of set "i".
       n=n+num_jts
       final(i)=n
c   "final(i)" is the last element of set "i".
       read (10,306) (alpha(k),beta(k),k=l,n)
       last=final(num)
    10 continue
       go to 65
c
c   OPTION 2:
    15 read (10,301) locality,sheet,county,quad,systemfm
c   Convert to ANSI Fortran 77 by changing to 'read (10,*) locality,' etc.
       read (10,302) lat,long,date
c   Convert to ANSI Fortran 77 by changing to 'read (10,*) lat,long,date'.
       read (10,310) stdr1(1),alpha(1),stdr2(1),beta(1),ddr(1),overt
       read (10,304) num
c   "num"=number of sets considered from locality
       n=0
       bed=alpha(1)+beta(1)
       if (bed.gt.0.) n=1
       do 20 i=1,num
       read (10,305) lithology(i),rank(i),num_jts
c   Convert to ANSI Fortran 77 by changing to 'read (10,*)' etc.
c   "num_jts" is the number of joints in set "i".
       init(i)=n+1
       l=init(i)
c   "init(i)" is the first element of set "i".
       n=n+num_jts
       final(i)=n
c   "final(i)" is the last element of set "i".
       read (10,315) (stdr1(k),alpha(k),stdr2(k),beta(k),ddr(k),k=l,n)
    20 continue
```

```
      last=final(num)
      do 50 k=1,last
      if (stdr1(k).eq.'S') go to 25
      if (stdr2(k).eq.'E') go to 35
      alpha(k)=360.-alpha(k)
      go to 40
   25 if (stdr2(k).eq.'W') go to 30
      alpha(k)=180.-alpha(k)
      go to 45
   30 alpha(k)=180.+alpha(k)
      if ((ddr(k).eq.' W').or.(ddr(k).eq.'NW').or.(ddr(k).eq.' N')) go
     2 to 50
      alpha(k)=alpha(k)-180.
      go to 50
   35 if ((ddr(k).eq.' E').or.(ddr(k).eq.'SE').or.(ddr(k).eq.' S')) go
     2 to 50
      alpha(k)=alpha(k)+180.
      go to 50
   40 if ((ddr(k).eq.' N').or.(ddr(k).eq.'NE').or.(ddr(k).eq.' E')) go
     2 to 50
      alpha(k)=alpha(k)-180.
   45 if ((ddr(k).eq.' S').or.(ddr(k).eq.'SW').or.(ddr(k).eq.' W')) go
     2 to 50
      alpha(k)=alpha(k)+180.
   50 continue
      if (bed.eq.0.) go to 65
      al=alpha(1)
      be=beta(1)
      if (overt.ne."O") go to 55
      al=al+180.
      be=180.-be
   55 do 60 i=2,last
      n=i-1
      alpha(n)=alpha(i)
   60 beta(n)=beta(i)
      do 62 in=1,num
      init(in)=init(in)-1
   62 final(in)=final(in)-1
      last=last-1
c
c
   65 read (10,307) corr,decl
c If no declination correction indicated, "corr"=0.; if declination correction
c is needed, "corr" is a positive number between 1. and 9. "decl" is positive
c if west, negative if east, and can be given to nearest tenth of degree.
      last=final(num)
      if(corr.eq.0.) go to 75
      do 70 i=1,last
   70 alpha(i)=alpha(i)-decl
   75 continue
c
```

```
c+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
c     INITIAL WRITE STATEMENTS, COMPUTATION OF POLES TO PLANES AND COMPUTATION
c     OF DIRECTION COSINES OF POLES TO PLANES:
c
      write (11,401) locality,lat,long,sheet,county,quad
      write (11,402) systemfm,date,num
      write (11,403) locality
      do 80 i=1,num
      j=init(i)
      k=final(i)
      write (11,404) i,rank(i),lithology(i),(alpha(l),beta(l),l=j,k)
   80 continue
      do 95 n=1,last
      alpha(n)=alpha(n)-90.
      if(alpha(n).lt.0.) alpha(n)=360.+alpha(n)
      if(beta(n))85,90,90
   85 beta(n)=-(90.+beta(n))
      go to 95
   90 beta(n)=90.-beta(n)
   95 continue
c
      write (11,405) locality
      do 100 i=1,num
      j=init(i)
      k=final(i)
      write (11,406) i,rank(i),lithology(i),(alpha(l),beta(l),l=j,k)
  100 continue
      do 105 n=1,last
      alpha(n)=alpha(n)*rad
  105 beta(n)=beta(n)*rad
      do 110 i=1,last
      c(1,i)=cos(alpha(i))*cos(beta(i))
      c(2,i)=sin(alpha(i))*cos(beta(i))
  110 c(3,i)=sin(beta(i))
      write (11,401) locality,lat,long,sheet,county,quad
      write (11,402) systemfm,date,num
```

```
c+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
c     DETERMINATION OF MEANS AND ANGLES OF ERROR FOR EACH NONTRIVIAL SET:
c
      no=0
      do 120 i=1,num
      j=init(i)
      k=final(i)
      n=k-j+1
      if(j.eq.k) go to 120
      no=no+1
      write (11,410) rank(i), lithology(i), n
      call mean(j,k,no,xm)
c  "xm" is the radius of confidence calculated by subroutine mean for the set.
c  "no" is the subscript value of the noth calculated mean vector.
c  "j,k" are respectively the initial and final subscript values of the set.
      if(xm.eq.0.) go to 120
      rdconf=acos(xm)/rad
      call angle(no)
      write (11,415) i,alpha(no),beta(no),xm,rdconf
      str=alpha(no)+90.
      if(str.gt.360.) str=str-360.
      dip=90-beta(no)
      write (11,420) i,str,dip
  120 continue
  300 format (i1)
c  Omit format statements 301,302, and 305, when converting to ANSI Fortran 77.
  301 format (v)
  302 format (v)
  303 format (f5.1,f4.0)
  304 format (i1)
  305 format (v)
  306 format (3(2f4.0,1x))
  307 format (f2.0,f5.1)
  310 format (a1,f4.1,a1,1x,f3.0,a2,1x,a1)
  315 format (3(a1,f3.0,a1,1x,f3.0,a2,1x))
  401 format (1h0,'Locality:',2x,a7,//,5x,'Latitude:',1x,a8,3x,'Longitud
     2e:',1x,a8,/,5x,'Sheet:',1x,a16,/,5x,'County:',1x,a16,/,5x,'Quad:',
     31x,a16)
  402 format (1h0,4x,'System/formation:',1x,a20,/,5x,'Date measured:',1x
     2,a8,//,5x,'Unrotated results from',1x,i1,1x,'sets:')
  403 format (1h0,'Strikes and dips of data planes:',2x,'Locality',1x,a7
     2,/,5x,'Strike azimuth is 90 degrees less than dip direction.')
  404 format (' ',//,2x,'Set no. ',i1,2x,a12,a12,/,8x,'strike',2x,'dip',
     2/,8x,'azimuth',1x,'in degrees.',/,(10x,2(f4.0,2x)))
  405 format (' ',//////,1x,a7,3x,'POLES TO DATA PLANES',/,5x,'(Pole dir
     2ection and dip in degrees)')
  406 format (' ',//,' Set no.',1x,i1,2x,a12,a12,/(10x,2(f4.0,2x)))
  410 format (' ',/,10x,a12,a12,'n=',i2)
  415 format (' ',4hSet ,i1,2x,'MEAN POLE and its RADIUS OF CONFIDENCE',
     2/,10x,'Azimuth',1x,f6.2,2x,'Dip in azimuth direction',1x,f5.2,/,10
     3x,'Radius of confidence',f8.4,/,10x,'Angle of error',f6.2,1x,'degr
     4ees',/,10x,'The probability is 5% that the angle of error',/,10x,'
     5will exceed this value (i.e., 95% confidence level).'/)
```

```
  420 format (' ',7x,'MEAN PLANE, set no. ',i1,/,10x'Strike:',f6.1,1x,'D
     2ip:',f6.1,/)
  425 format (1h1,'ROTATED DATA:',2x,a7,/, 7x,i1,1x,'sets, rotated to be
     2dding flat with respect to:',/, 7x,f7.1,1x,'Azimuth,',f4.0,1x,'Dip
     3 of bedding')
  430 format (' ',//,' SET No.',i1,2x,a12,1x,'set of fractures in',1x,a1
     22,2x,'n=',i2,/)
  435 format (' ',11x,'AZIMUTH AND DIP OF MEAN POLE',/,11x,f7.1,2x,f7.1)
  440 format (' ',11x,'Azimuth and dip of rotated poles to planes',/(10x
     2,f7.1,2x,f7.1))
c
c+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
c     DETERMINATION OF ROTATION MATRIX:
c
      if(be.eq.0.0) go to 190
      if(corr.eq.0.) go to 125
c declination correction for strike of bedding if needed
      al=al-decl
  125 al=al*rad
      be=be*rad
c  "u" is axis of rotation and strike of bedding
      u(1)=cos(al)
      u(2)=sin(al)
      u(3)=0.0
      u11=u(1)**2
      u22=u(2)**2
c  "q" is unit vector perpendicular to "u", the axis of rotation, which
c  determines the angle of rotation. If below horizon to the west by angle
c  "be", the coordinate system is rotated counterclockwise from north by "be"
c  degrees. If below horizon in east by angle "be" the coordinate system
c  is rotated clockwise "be" degrees.
      q(1)=-sin(be)*u(2)
      q(2)=sin(be)*u(1)
      q(3)=cos(be)
      a(1,1)=u11+q(3)*u22
      a(2,2)=u22+q(3)*u11
      a(1,2)=u(1)*u(2)*(1-q(3))
      a(2,1)=a(1,2)
      do 130 i=1,3
  130 a(i,3)=q(i)
      a(3,1)=-a(1,3)
      a(3,2)=-a(2,3)
c     a(i,j) is the rotation matrix
      do 145 i=1,3
      r=0.0
      do 135 j=1,3
  135 r=r+a(i,j)**2
      r=sqrt(r)
      do 140 j=1,3
  140 a(i,j)=a(i,j)/r
  145 continue
```

```
c+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
c     ROTATION OF DATA TO BEDDING FLAT:
c
      do 155 i=1,last
      n=no+i
      do 150 j=1,3
  150 cm(n,j)=c(j,i)
  155 continue
      last=last+no
      do 175 i=1,last
      do 165 j=1,3
      sum=0.0
      do 160 l=1,3
  160 sum=sum+a(j,l)*cm(i,l)
  165 prod(j)=sum
      do 170 l=1,3
  170 cm(i,l)=prod(l)
      call unvec(i)
      call angle(i)
  175 continue
      alr=al/rad
      ber=be/rad
c
c+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
c     FINAL SORTING AND FINAL WRITE STATEMENTS:
c
      write (11,425) locality,num,alr,ber
      it=0
      do 185 i=1,num
      j=init(i)+no
      k=final(i)+no
      n=k-j+1
      write (11,430) i,rank(i),lithology(i),n
      if(j.eq.k) go to 180
      it=it+1
      write (11,435) alpha(it),beta(it)
      str=alpha(it)+90.
      if(str.gt.360.) str=str-360.
      dip=90-beta(it)
      write (11,420) i,str,dip
  180 write (11,440) (alpha(l),beta(l),l=j,k)
  185 continue
  190 read (10,320) lum
  320 format (i1)
      if(lum.gt.0) go to 5
c PROGRAM WILL REPEAT FOR "lum.gt.0" FOR AS MANY LOCALITIES AS NEEDED:
      end
```

```
c***************************************************************************
c                        SUBROUTINE MEAN
c
c     This subroutine calculates the mean and radius of confidence of sets
c     of unit vectors in array "c(i,60)".
c
      subroutine mean (j,k,no,xs)
c  "j" and "k" define the range of "c" to be considered.
      dimension cs(3)
      common c(3,60),cm(70,3)
      double precision r,cs
      x=k-j+1
      do 30 l=1,3
      r=0.0
      do 20 i=j,k
   20 r=r+c(l,i)
      cm(no,l)=r
   30 cs(l)=r**2
      r=sqrt(cs(1)+cs(2)+cs(3))
c  'r' is now the length of the vector sum.
      do 50 l=1,3
   50 cm(no,l)=cm(no,l)/r
c  'cm' is now the vector mean (the mean pole).
      ex=x-2
      if(r.gt.ex) go to 60
      write (11,400) j,k
  400 format (1h0,10x,'Set of data including planes no.',1x,i2,1x,'throu
     2gh no.',1x,i2,1x,'cannot be treated accurately by this program',/,
     310x,'as r is less than or equal to the number of planes minus 2 (n
     4-2).',/,10x,'A valid measure of dispersion is not computed.',//)
      xs=0.
      go to 70
   60 g=1./(x-1.)
      p=0.05
      xs=1.-((x-r)*((1./p)**g-1.))/r
   70 return
      end
```

```
c*************************************************************************
c                         SUBROUTINE ANGLE
c
c     This subroutine converts direction cosine array to azimuth and plunge.
c
      subroutine angle (i)
      common p(180),cm(70,3),alpha(70),beta(70),rad
      abort=cm(i,1)+cm(i,2)+cm(i,3)
      if (abort.eq.0.0) go to 20
      if(cm(i,3)) 4,8,8
    4 cm(i,1)=-cm(i,1)
      cm(i,2)=-cm(i,2)
      cm(i,3)=-cm(i,3)
    8 if(cm(i,3).eq.1.) go to 18
      cm(i,1) = cm(i,1)/sqrt(cm(i,1)**2+cm(i,2)**2)
      alpha(i)=acos(cm(i,1))/rad
      b=cm(i,3)/sqrt(1.-cm(i,3)**2)
      beta(i)=atan(b)/rad
      if (cm(i,1)) 10,14,14
   10 if(cm(i,2)) 12,16,16
   12 alpha(i)=360.-abs(alpha(i))
      go to 20
   14 if(cm(i,2).gt.0) go to 20
      alpha(i)=360.-alpha(i)
      go to 20
   16 if(alpha(i).gt.90.) go to 20
      alpha(i)=180.-abs(alpha(i))
      go to 20
   18 alpha(i)=0.
      beta(i)=90.
   20 return
      end
c
c*************************************************************************
c                      SUBROUTINE UNVEC
c
c     This subroutine verifies that unit vector array "cm(n,i)" is unitized.
c
      subroutine unvec(j)
      common p(180),cm(70,3)
      r=0.0
      do 10 i=1,3
   10 r=r+cm(j,i)**2
      if(r.eq.0.) go to 20
      r=sqrt(r)
      do 15 i=1,3
   15 cm(j,i)=cm(j,i)/r
   20 return
      end
c
c*************************************************************************
c                        END OF SOURCE CODE
```

REFERENCE

Fisher, R., 1953, Dispersion on a sphere: Proc. Royal Soc., pt. A, v. 217, p. 295-305.

APPENDIX

Sample of data package, input option 1:

```
1
'Va_2a' 'Bluefield_2d' 'Giles' 'Narrows_7_1/2' 'Silurian/Rose_Hill'
'37d20.9m' '80d48.5m' '4/27/78'
 46.5 18.
6
'ss-slts' 'sys(1)' 18
173. 83. 169. 87. 170. 87.
178. 81. 174. 78. 170. 83.
172. 86. 177. 79. 185. 87.
173. 83. 180. 83. 179. 85.
175. 80. 175. 92. 175. 81.
175. 78. 176. 86. 177. 87.
'ss/slts' 'rank(2)' 16
289. 73. 290. 75. 290. 73.
293. 72. 288. 72. 288. 93.
300. 87. 295. 73. 284. 73.
288. 76. 285. 81. 284. 77.
298. 81. 288. 75. 289. 80.
293. 76.
'ss/slts' 'rank(3)' 11
235. 76. 245. 70. 237. 72.
252. 74. 246. 75. 235. 96.
216. 67. 246. 85. 229. 71.
260. 88. 246. 70.
'Shale' 'sys(1)' 2
335. 92. 348. 86.
'Shale' 'rank(2)' 1
108. 87.
'Shale' 'rank(3)' 2
244. 81. 229. 70.
1.  3.5
0
```

The first line of data gives the option number in column 1. The second line names the locality, sheet, county, quadrangle, and system/formation. The third line gives the latitude, longitude, and the date that the data were acquired. The fourth line gives the azimuth of strike and the dip of bedding using the conventions described in comments in the source code. The fifth line gives the number of data sets input from the locality. The sixth line gives the lithology, rank, and number of measurements in set 1. The seventh and similar succeeding lines give the azimuth of the strike and the dip of the data planes using conventions of option 1 in the source code. The next to last line contains a 1 in column 1, indicating that a declination correction is required. If the correction is to the east, a minus is entered in column 3. The correction is given in columns 3 through 7 in the form ±__.__ degrees. The last line contains a 0 in column 1 indicating that no other localities will be considered in this run. Otherwise a 1 through 9 will be entered, followed by data from the next locality entered as that above. The data are printed out in the same order as entered.

Sample of data output, option 1:

Locality: Va_2a

Latitude: 37d20.9m Longitude: 80d48.5m
Sheet: Bluefield_2d
County: Giles
Quad: Narrows_7_1/2

System/formation: Silurian/Rose_Hill
Date measured: 4/27/78

Unrotated results from 6 sets:

Strikes and dips of data planes: Locality Va_2a
Strike azimuth is 90 degrees less than dip direction.

```
 Set no. 1  sys(1)      ss-slts
      strike  dip
      azimuth in degrees.
        170.   83.
        166.   87.
        167.   87.
        175.   81.
        171.   78.
        167.   83.
        169.   86.
        174.   79.
        182.   87.
        170.   83.
        177.   83.
        176.   85.
        172.   80.
        172.   92.
        172.   81.
        172.   78.
        173.   86.
        174.   87.
```

Set no. 2 rank(2) ss/slts

strike azimuth	dip in degrees.
286.	73.
287.	75.
287.	73.
290.	72.
285.	72.
285.	93.
297.	87.
292.	73.
281.	73.
285.	76.
282.	81.
281.	77.
295.	81.
285.	75.
286.	80.
290.	76.

Set no. 3 rank(3) ss/slts

strike azimuth	dip in degrees.
232.	76.
242.	70.
234.	72.
249.	74.
243.	75.
232.	96.
213.	67.
243.	85.
226.	71.
257.	88.
243.	70.

Set no. 4 sys(1) Shale

strike azimuth	dip in degrees.
332.	92.
345.	86.

Set no. 5 rank(2) Shale

strike azimuth	dip in degrees.
105.	87.

Set no. 6 rank(3) Shale

strike	dip
azimuth in degrees.	
241.	81.
226.	70.

Va_2a POLES TO DATA PLANES
(Pole direction and dip in degrees)

Set no. 1 sys(1) ss-slts

80.	7.
76.	3.
77.	3.
85.	9.
81.	12.
77.	7.
79.	4.
84.	11.
92.	3.
80.	7.
87.	7.
86.	5.
82.	10.
82.	-2.
82.	9.
82.	12.
83.	4.
84.	3.

Set no. 2 rank(2) ss/slts

196.	17.
197.	15.
197.	17.
200.	18.
195.	18.
195.	-3.
207.	3.
202.	17.
191.	17.
195.	14.
192.	9.
191.	13.
205.	9.
195.	15.
196.	10.
200.	14.

Set no. 3 rank(3)   ss/slts

142.	14.
152.	20.
144.	18.
159.	16.
153.	15.
142.	-6.
123.	23.
153.	5.
136.	19.
167.	2.
153.	20.

Set no. 4 sys(1)   Shale

242.	-2.
255.	4.

Set no. 5 rank(2)   Shale

15.	3.

Set no. 6 rank(3)   Shale

151.	9.
136.	20.

Locality: Va_2a

Latitude: 37d20.9m   Longitude: 80d48.5m
Sheet: Bluefield_2d
County: Giles
Quad: Narrows_7_1/2

System/formation: Silurian/Rose_Hill
Date measured: 4/27/78

Unrotated results from 6 sets:

sys(1)   ss-slts   n=18
Set 1 MEAN POLE and its RADIUS OF CONFIDENCE
Azimuth 81.66 Dip in azimuth direction 6.35
Radius of confidence 0.9992
Angle of error 2.32 degrees
The probability is 5% that the angle of error will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 1
Strike: 171.7 Dip: 83.7

rank(2) ss/slts n=16

Set 2 MEAN POLE and its RADIUS OF CONFIDENCE

Azimuth 196.64 Dip in azimuth direction 12.74
Radius of confidence 0.9983
Angle of error 3.39 degrees
The probability is 5% that the angle of error will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 2
Strike: 286.6 Dip: 77.3

rank(3) ss/slts n=11

Set 3 MEAN POLE and its RADIUS OF CONFIDENCE

Azimuth 147.33 Dip in azimuth direction 13.56
Radius of confidence 0.9894
Angle of error 8.35 degrees
The probability is 5% that the angle of error will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 3
Strike: 237.3 Dip: 76.4

sys(1) Shale n= 2

Set 4 MEAN POLE and its RADIUS OF CONFIDENCE

Azimuth 247.99 Dip in azimuth direction 1.01
Radius of confidence 0.8509
Angle of error 31.69 degrees
The probability is 5% that the angle of error will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 4
Strike: 338.0 Dip: 89.0

rank(3) Shale n= 2

Set 6 MEAN POLE and its RADIUS OF CONFIDENCE

Azimuth 143.19 Dip in azimuth direction 14.62
Radius of confidence 0.7580
Angle of error 40.72 degrees
The probability is 5% that the angle of error will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 6
Strike: 233.2 Dip: 75.4

ROTATED DATA: Va_2a
6 sets, rotated to bedding flat with respect to:
43.0 Azimuth, 18. Dip of bedding.

SET No.1 sys(1) set of fractures in ss-slts n=18

AZIMUTH AND DIP OF MEAN POLE
261.8 5.0
MEAN PLANE, set no. 1
Strike: 351.8 Dip: 85.0

Azimuth and dip of rotated poles to planes

259.9	3.8
255.0	6.7
256.0	6.9
265.2	3.1
82.1	0.8
257.0	3.1
258.2	6.5
264.7	0.9
270.7	10.4
259.9	3.8
266.7	5.5
265.2	7.2
262.6	1.4
259.6	13.0
262.3	2.4
83.0	0.5
262.1	7.5
262.8	8.7

SET No.2 rank(2) set of fractures in ss/slts n=16

AZIMUTH AND DIP OF MEAN POLE
193.9 4.0
MEAN PLANE, set no. 2
Strike: 283.9 Dip: 86.0

Azimuth and dip of rotated poles to planes

192.0	8.1
193.5	6.5
192.9	8.4
195.4	10.2
190.8	8.8
16.5	11.4
26.4	2.2
197.5	9.8
187.3	6.9
191.9	5.0
10.4	0.6
188.4	3.0
202.7	3.0
191.6	6.0
193.9	1.4
196.5	6.3

SET No.3 rank(3) set of fractures in ss/slts n=11

AZIMUTH AND DIP OF MEAN POLE
327.7 4.1
MEAN PLANE, set no. 3
Strike: 57.7 Dip: 85.9

Azimuth and dip of rotated poles to planes

321.3	3.8
150.4	2.9
143.0	0.3
337.4	0.3
331.8	2.0
322.2	23.8
123.3	5.3
332.9	12.0
135.4	1.0
347.5	13.0
151.3	3.0

SET No.4 sys(1) set of fractures in Shale n= 2

AZIMUTH AND DIP OF MEAN POLE
246.6 8.5
MEAN PLANE, set no. 4
Strike: 336.6 Dip: 81.5

Azimuth and dip of rotated poles to planes

241.2	3.7
252.1	13.1

SET No.5 rank(2) set of fractures in Shale n= 1

Azimuth and dip of rotated poles to planes

16.5	11.4

SET No.6 rank(3) set of fractures in Shale n= 2

AZIMUTH AND DIP OF MEAN POLE

323.8	3.4

MEAN PLANE, set no. 6

Strike: 53.8 Dip: 86.6

Azimuth and dip of rotated poles to planes

330.5	8.2
135.4	2.0

Sample of data package, input option 2:

```
2
'W.Va-18' 'Bluefield_2d' 'Greenbrier' 'Jerrys_Run_7.5' 'Devonian/Foreknobs'
'37d47.5m' '80d14m' '7/28/77'
N 1.5E 26.SE
3
'ss/mslts/sh' 'sys(1)' 14
N82.E 88.NW N76.E 88.NW N71.E 89.NW
N77.E 89.NW N73.E 86.NW N83.E 90.NW
N73.E 89.NW N72.E 86.NW N70.E 68.NW
N76.E 71.NW N75.E 86.NW N80.E 90.NW
N90.E 91.NW N90.E 90.NW
'mudstone' 'rank(2)' 10
N43.W 61.SW N24.W 76.SW N39.W 81.SW
N18.W 76.SW N34.W 79.SW N33.W 77.SW
N49.W 78.SW N54.W 72.SW N38.W 69.SW
N24.W 74.SW
'Sandstone' 'rank(2)' 4
N22.E 86.NW N11.E 70.NW N13.E 74.NW
N20.E 82.NW
0.
0
```

In this option strike directions are recorded in conventional quadrant form for easier entry. It is extremely important that the character data relating to strike and dip be entered in the proper column, i.e., N in column 1, E in column 5, etc. If decimal points are omitted, numbers need to be right justified. 82_ instead of 82. will be read as 820. This is also the case for option 1.

Sample of data output, option 2:

Locality: W.Va-18

Latitude: 37d47.5m   Longitude: 80d14m
Sheet: Bluefield_2d
County: Greenbrier
Quad: Jerrys_Run_7.5

System/formation: Devonian/Foreknobs
Date measured: 7/28/77

Unrotated results from 3 sets:

Strikes and dips of data planes: Locality W.Va-18
Strike azimuth is 90 degrees less than dip direction.

Set no. 1 sys(1) ss/mslts/sh

strike azimuth	dip in degrees.
262.	88.
256.	88.
251.	89.
257.	89.
253.	86.
263.	90.
253.	89.
252.	86.
250.	68.
256.	71.
255.	86.
260.	90.
270.	91.
270.	90.

Set no. 2 rank(2) mudstone

strike azimuth	dip in degrees.
137.	61.
156.	76.
141.	81.
162.	76.
146.	79.
147.	77.
131.	78.
126.	72.
142.	69.
156.	74.

Set no. 3 rank(2) Sandstone

strike	dip
azimuth in degrees.	
202.	86.
191.	70.
193.	74.
200.	82.

W.Va-18 POLES TO DATA PLANES
(Pole direction and dip in degrees)

Set no. 1 sys(1) ss/mslts/sh

172.	2.
166.	2.
161.	1.
167.	1.
163.	4.
173.	0.
163.	1.
162.	4.
160.	22.
166.	19.
165.	4.
170.	0.
180.	-1.
180.	0.

Set no. 2 rank(2) mudstone

47.	29.
66.	14.
51.	9.
72.	14.
56.	11.
57.	13.
41.	12.
36.	18.
52.	21.
66.	16.

Set no. 3 rank(2) Sandstone

112.	4.
101.	20.
103.	16.
110.	8.

Locality: W.Va-18

Latitude: 37d47.5m   Longitude: 80d14m
Sheet: Bluefield_2d
County: Greenbrier
Quad: Jerrys_Run_7.5

System/formation: Devonian/Foreknobs
Date measured: 7/28/77

Unrotated results from 3 sets:

sys(1)   ss/mslts/sh n=14
Set 1 MEAN POLE and its RADIUS OF CONFIDENCE
Azimuth 167.75 Dip in azimuth direction 4.21
Radius of confidence 0.9966
Angle of error 4.73 degrees
The probability is 5% that the angle of error
will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 1
Strike: 257.8 Dip: 85.8

rank(2)   mudstone   n=10
Set 2 MEAN POLE and its RADIUS OF CONFIDENCE
Azimuth 54.50 Dip in azimuth direction 15.96
Radius of confidence 0.9915
Angle of error 7.48 degrees
The probability is 5% that the angle of error
will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 2
Strike: 144.5 Dip: 74.0

rank(2)   Sandstone   n= 4
Set 3 MEAN POLE and its RADIUS OF CONFIDENCE
Azimuth 106.61 Dip in azimuth direction 12.04
Radius of confidence 0.9842
Angle of error 10.21 degrees
The probability is 5% that the angle of error
will exceed this value (i.e., 95% confidence level).

MEAN PLANE, set no. 3
Strike: 196.6 Dip: 78.0

ROTATED DATA: W.Va-18

3 sets, rotated to bedding flat with respect to:

1.5 Azimuth, 26. Dip of bedding.

SET No.1 sys(1) set of fractures in ss/mslts/sh n=14

AZIMUTH AND DIP OF MEAN POLE

347.3	2.2

MEAN PLANE, set no. 1

Strike: 77.3 Dip: 87.8

Azimuth and dip of rotated poles to planes

352.1	2.3
346.7	4.9
342.5	7.9
348.0	5.4
343.1	4.4
353.8	3.7
344.3	7.1
342.2	4.8
152.9	10.8
159.4	10.5
344.9	3.5
351.1	5.0
0.6	1.6
0.2	0.7

SET No.2 rank(2) set of fractures in mudstone n=10

AZIMUTH AND DIP OF MEAN POLE

234.9	5.1

MEAN PLANE, set no. 2

Strike: 324.9 Dip: 84.9

Azimuth and dip of rotated poles to planes

53.1	9.3
246.4	9.6
230.7	10.9
252.3	10.6
236.1	10.3
237.6	8.6
222.3	4.9
39.8	2.4
55.1	0.4
246.8	7.6

SET No.3 rank(2) set of fractures in Sandstone n= 4

AZIMUTH AND DIP OF MEAN POLE

287.0	13.1

MEAN PLANE, set no. 3

Strike: 17.0 Dip: 76.9

Azimuth and dip of rotated poles to planes

293.4	20.3
280.5	5.7
282.7	9.5
290.6	16.7